

LDL06E04A1

2008 年 9 月 28 日

第 1 章

論理譜

```

{ ===== }
{ キーボード 4 行 4 列 }
{ ===== }
logicname LDL06E04A1

{ ===== }
{ 実効譜 }
{ ===== }
entity main
{ ----- }
{ 入力 }
{ ----- }
input RESET; { 初期化 }
input SCLK; { 駆動 CLK }
input YIN[4]; { 行入力 }
input TAKE; { データ読み取り }
{ ----- }
{ 出力 }
{ ----- }
output Q[4]; { データ }
output SET; { データ確定 }
output XSEL[4]; { 列出力 }
output OFF;
{ ----- }
{ 内部信号 }
{ ----- }
bitr sclkp[2]; { SCLK 行程 }
bitr data[4]; { データ }
bitr count[2]; { 駆動計数 }
bitr set[3]; { 主行程 }
bitn keypush;
bitn off;

{ ----- }
{ 検査端子 }
{ ----- }

output TOP[3]; TOP=set;
output T1P[2]; T1P=sclkp;
output T2P[4]; T2P=data;
output T3P[2]; T3P=count;
output T4P; T4P=keypush;
output T5P; T5P=off;

{ ----- }
{ 出力代入 }
{ ----- }

```

```

{ ----- }
Q=data;
OFF=off;

switch(set)
  case 1: SET=1;
endswitch

{ ----- }
{ SCLK 行程 }
{ ----- }

if (RESET)
  sclkp=0;
else
  if (SCLK)
    switch(sclkp)
      case 0: sclkp=1;
      case 1: sclkp=2;
      default: sclkp=sclkp;
    endswitch
  else
    sclkp=0;
  endif
endif

{ ----- }
{ 駆動計数 }
{ ----- }

if (RESET)
  count=0;
else
  if (sclkp.0)
    if (count==3)
      count=0;
    else
      count=count+1;
    endif
  else
    count=count;
  endif
endif

{ ----- }
{ 列駆動 }
{ ----- }

switch(count)
  case 0: XSEL=0b0001;
  case 1: XSEL=0b0010;
  case 2: XSEL=0b0100;
  case 3: XSEL=0b1000;
endswitch

{ ----- }
{ キー離し検出 }
{ ----- }

switch(set)
  case 0: off=1;
endswitch

{ ----- }
{ 主行程 }
{ ----- }

if (RESET)
  set=0;
else

```

```

switch(set)
case 0:
    if (sclkp.0)
        if (keypush)
            set=1;
        endif
    endif
case 1:
    if (TAKE)
        set=2;
    else
        set=set;
    endif
case 2:
    switch(count)
        case 0:
            switch(YIN)
                case 0: set=3;
                default: set=set;
            endswitch
        default: set=set;
    endswitch
case 3:
    switch(count)
        case 1:
            switch(YIN)
                case 0: set=4;
                default: set=2;
            endswitch
        default: set=set;
    endswitch
case 4:
    switch(count)
        case 2:
            switch(YIN)
                case 0: set=5;
                default: set=2;
            endswitch
        default: set=set;
    endswitch
case 5:
    switch(count)
        case 3:
            switch(YIN)
                case 0: set=0;
                default: set=2;
            endswitch
        default: set=set;
    endswitch
endswitch
endif

{ ----- }
{ キー押し下げ検出 }
{ ----- }

if ((YIN!=0)&sclkp.0)
    switch(set)
        case 0: keypush=1;
    endswitch
endif

{ ----- }
{ 行データ }
{ ----- }

if (RESET)
    data.0:1=0;

```

```
else
  if (keypush)
    if (YIN.0)
      data.0:1=0;
    else
      if (YIN.1)
        data.0:1=1;
      else
        if (YIN.2)
          data.0:1=2;
        else
          if (YIN.3)
            data.0:1=3;
          endif
        endif
      endif
    endif
  else
    data.0:1=data.0:1;
  endif
endif

{ ----- }
{ 列データ }
{ ----- }

if (RESET)
  data.2:3=0;
else
  if (keypush)
    data.2:3=count;
  else
    data.2:3=data.2:3;
  endif
endif

ende
```

```

{ ===== }
{ 機能実行譜 }
{ ===== }
entity sim
{ ----- }
{ 出力 }
{ ----- }
output RESET;
output SCLK;
output YIN[4];
output TAKE;
output Q[4];
output SET;
output XSEL[4];
output OFF;
input simres;

{ ----- }
{ 内部信号 }
{ ----- }
bitr tc[10];
bitn node_XSEL[4];
bitn node_SET;
bitn node_RESET;
bitn node_TAKE;
bitn node_Q[4];
bitn node_OFF;
bitr tp[8];
bitr takep[2];
bitr chk;

{ ----- }
{ 検査端子 }
{ ----- }
output TBOP[8]; TBOP=tp;
output TB1P; TB1P=chk;

{ ----- }
{ 実効譜導入 }
{ ----- }
part main(node_RESET,SCLK,YIN,node_TAKE,node_Q,node_SET,node_XSEL,node_OFF)

RESET=node_RESET;
XSEL=node_XSEL;
SET=node_SET;
TAKE=node_TAKE;
Q=node_Q;
OFF=node_OFF;

{ ----- }
{ 検査位置 }
{ ----- }
simres=0;
if (!simres) tc=tc+1; endif

{ ----- }
{ 初期化 }
{ ----- }
if (tc<5) node_RESET=1; endif

{ ----- }
{ 駆動 CLK }
{ ----- }

```

```

SCLK=tc.2;

{ ----- }
{ データ読み取り }
{ ----- }
if (node_SET) node_TAKE=1; endif

{ ----- }
{ データ読み取り起点 }
{ ----- }
if (node_RESET)
  takep=0;
else
  switch(takep)
    case 0: if (node_OFF) takep=1; endif
    case 1: takep=2;
    case 2:
      if (node_OFF)
        takep=takep;
      else
        takep=0;
      endif
    endif
  endswitch
endif

{ ----- }
{ 検査 }
{ ----- }
if (node_RESET)
  chk=0;
else
  if (takep.0)
    if (tp.0:3==node_Q) chk=chk; else chk=1; endif
  else
    chk=chk;
  endif
endif

{ ----- }
{ キー押し下げ回数 }
{ ----- }
if (node_RESET)
  tp=0;
else
  if (takep==1)
    tp=tp+1;
  else
    tp=tp;
  endif
endif

{ ----- }
{ 行入力 }
{ ----- }
if (tc>10)
  if (node_OFF)
    switch(tp)
      case 0:
        switch(node_XSEL)
          case 1: YIN.0=1; { 0 }
        endswitch
      case 1:
        switch(node_XSEL)
          case 1: YIN.1=1; { 1 }
        endswitch
    endswitch
  endif
endif

```

```

case 2:
    switch(node_XSEL)
        case 1: YIN.2=1;    { 4 }
    endswitch
case 3:
    switch(node_XSEL)
        case 1: YIN.3=1;    { 7 }
    endswitch
case 4:
    switch(node_XSEL)
        case 2: YIN.0=1;    { = }
        YIN.1=1;
    endswitch
case 5:
    switch(node_XSEL)
        case 2: YIN.1=1;    { 2 }
    endswitch
case 6:
    switch(node_XSEL)
        case 2: YIN.2=1;    { 5 }
    endswitch
case 7:
    switch(node_XSEL)
        case 2: YIN.3=1;    { 8 }
    endswitch
case 8:
    switch(node_XSEL)
        case 4: YIN.0=1;    { - }
    endswitch
case 9:
    switch(node_XSEL)
        case 4: YIN.1=1;    { 3 }
    endswitch
case 10:
    switch(node_XSEL)
        case 4: YIN.2=1;    { 6 }
    endswitch
case 11:
    switch(node_XSEL)
        case 4: YIN.3=1;    { 9 }
    endswitch
case 12:
    switch(node_XSEL)
        case 8: YIN.0=1;    { C }
    endswitch
case 13:
    switch(node_XSEL)
        case 8: YIN.1=1;    { + }
    endswitch
case 14:
    switch(node_XSEL)
        case 8: YIN.2=1;    { / }
    endswitch
case 15:
    switch(node_XSEL)
        case 8: YIN.3=1;    { * }
    endswitch
endswitch
endif
endif

ende

endlogic

```