

LDL06E21A

2008 年 10 月 12 日

第 1 章

論理譜

```

{ ===== }
{ BCD2 進数変換器 }
{ ===== }
logicname LDL06E21A

{ ===== }
{ 実効譜 }
{ ===== }
entity main
{ ----- }
{ 入力 }
{ ----- }
input RESET;
input DATA[4];

{ ----- }
{ 出力 }
{ ----- }
output DIGITNO[3];
output Q[32];
output EOC;

{ ----- }
{ 内部信号 }
{ ----- }
bitr digitno[4];
bitn tmp[32];
bitr acc[32];
bitr seqno;
bitn tmp2shift[32];
bitn tmp3shift[32];
bitn digitno7p;

{ ----- }
{ 検査端子 }
{ ----- }
output TOP; TOP=seqno;
output T1P[32]; T1P=tmp2shift;
output T2P[32]; T2P=tmp3shift;

{ ----- }
{ 出力端子 }
{ ----- }
Q=acc;
DIGITNO=digitno.0:2;
EOC=digitno7p;

{ ----- }

```

```

{   BCD 値桁指定                               }
{-----}
if (RESET)
    digitno=7;
else
    if (digitno7p)
        digitno=digitno;
    else
        if (seqno==1)
            digitno=digitno-1;
        else
            digitno=digitno;
        endif
    endif
endif

{-----}
{   BCD 値桁指定 15 指標                               }
{-----}
digitno7p=digitno==15;

{-----}
{   BCD 値入力ビット変換                               }
{-----}
tmp.0:3=DATA;

{-----}
{   主行程                               }
{-----}
if (RESET)
    seqno=0;
else
    switch(seqno)
        case 0: seqno=1;
        case 1: seqno=0;
    endswitch
endif

{-----}
{   累算値 8 倍                               }
{-----}
tmp3shift.3:31=acc.0:28;

{-----}
{   累算値 2 倍                               }
{-----}
tmp2shift.1:31=acc.0:30;

{-----}
{   累算器                               }
{-----}
if (RESET)
    acc=0;
else
    if (digitno7p)
        acc=acc;
    else
        switch(seqno)
            case 0: acc=tmp3shift+tmp2shift;
            case 1: acc=acc+tmp;
        endswitch
    endif
endif
ende

```



```

{ ===== }
{ 機能実行譜 }
{ ===== }
entity sim
{ ----- }
{ 検査端子 }
{ ----- }
output RESET;
output DATA[4];
output DIGITNO[3];
output Q[32];
output EOC;
input simres;

{ ----- }
{ 内部信号 }
{ ----- }
bitr tc[8];
bitn node_DIGITNO[3];

{ ----- }
{ 出力代入 }
{ ----- }
DIGITNO=node_DIGITNO;

{ ----- }
{ 実効譜導入 }
{ ----- }
part main(RESET,DATA,node_DIGITNO,Q,EOC)

{ ----- }
{ 検査計数 }
{ ----- }
if (!simres) tc=tc+1; endif

{ ----- }
{ 初期化 }
{ ----- }
if (tc<5) RESET=1; endif

{ ----- }
{ BCD 値 }
{ ----- }
switch(node_DIGITNO)
case 7: DATA=1;
case 6: DATA=2;
case 5: DATA=3;
case 4: DATA=4;
case 3: DATA=5;
case 2: DATA=6;
case 1: DATA=7;
case 0: DATA=8;
endswitch

ende

endlogic

```